

MACHINE Wheels999

REFINES Display999

SEES ValMax999, Boolconv

VARIABLES

Ru

Rd

Rc

ud retenue unités dizaines

dc retenue dizaines centaines

reset

INVARIANTS

inv1 : $Ru \in 0 \dots 9$

inv2 : $Rd \in 0 \dots 9$

inv3 : $Rc \in 0 \dots 9$

inv4 : $ud \in 0 \dots 1$

inv5 : $dc \in 0 \dots 1$

inv6 : $reset \in \text{BOOL}$

inv7 : $reset = \text{FALSE} \wedge ud = 0 \wedge dc = 0 \Rightarrow u = Ru \wedge d = Rd \wedge c = Rc$

inv8 : $ud = 0 \vee dc = 0$

inv9 : $reset = \text{FALSE} \wedge ud = 1 \Rightarrow Ru = 0 \wedge u = 9 \wedge d = Rd \wedge c = Rc$

inv10 : $reset = \text{FALSE} \wedge dc = 1 \Rightarrow Ru = 0 \wedge u = 9 \wedge Rd = 0 \wedge d = 9 \wedge c = Rc$

inv11 : $reset = \text{TRUE} \Rightarrow u > 0 \vee d > 0 \vee c > 0$

inv12 : $reset = \text{TRUE} \Rightarrow dc = 0 \wedge ud = 0$

dlf1 : $reset = \text{TRUE} \Rightarrow (Ru = 0 \wedge Rd = 0 \wedge Rc = 0) \vee Ru > 0 \vee Rd > 0 \vee Rc > 0$

dlf2 : $reset = \text{FALSE} \Rightarrow (ud = 0 \wedge dc = 0 \wedge Ru < 9) \vee$
 $(ud = 0 \wedge Ru = 9) \vee (ud = 1 \wedge Rd < 9) \vee$
 $(ud = 1 \wedge dc = 0 \wedge Rd = 9) \vee (dc = 1 \wedge Rc = 9) \vee$
 $(dc = 1 \wedge Rc < 9) \vee$
 $((Ru > 0 \vee Rd > 0 \vee Rc > 0) \wedge dc = 0 \wedge ud = 0)$

EVENTS

Initialisation

begin

act5 : $Ru, Rd, Rc := 0, 0, 0$

act6 : $ud, dc := 0, 0$

act7 : $reset := \text{FALSE}$

end

Event *incru* $\hat{=}$

refines *incru*

when

grd5 : $reset = \text{FALSE}$

grd4 : $dc = 0$

grd1 : $Ru < 9$

grd6 : $ud = 0$

then

act1 : $Ru := Ru + 1$

end

Event *u9_0* $\hat{=}$

```

Status convergent
  when
    grd1 :  $Ru = 9$ 
    grd2 :  $reset = FALSE$ 
    grd3 :  $ud = 0$ 
  then
    act1 :  $Ru, ud := 0, 1$ 
  end
Event  $incrd \hat{=}$ 
refines  $incrd$ 
  when
    grd1 :  $ud = 1$ 
    grd2 :  $reset = FALSE$ 
    grd3 :  $Rd < 9$ 
  then
    act1 :  $Rd := Rd + 1$ 
    act2 :  $ud := 0$ 
  end
Event  $incrc \hat{=}$ 
refines  $incrc$ 
  when
    grd1 :  $dc = 1$ 
    grd2 :  $reset = FALSE$ 
    grd3 :  $Rc < 9$ 
  then
    act1 :  $Rc := Rc + 1$ 
    act2 :  $dc := 0$ 
  end
Event  $resetU \hat{=}$ 
Status convergent
  when
    grd3 :  $reset = TRUE$ 
    grd2 :  $Ru > 0$ 
  then
    act2 :  $Ru := Ru - 1$ 
  end
Event  $d9\_0 \hat{=}$ 
Status convergent
  when
    grd1 :  $ud = 1$ 
    grd2 :  $dc = 0$ 
    grd3 :  $reset = FALSE$ 
    grd4 :  $Rd = 9$ 
  then
    act1 :  $ud, dc := 0, 1$ 
    act2 :  $Rd := 0$ 
  end
Event  $c9\_0 \hat{=}$ 
refines  $reset$ 
  when
    grd1 :  $dc = 1$ 

```

```

        grd2 : Rc = 9
        grd3 : reset = FALSE
    then
        act1 : Rc := 0
        act2 : dc := 0
    end
Event resetD ≐
Status convergent
when
    grd1 : reset = TRUE
    grd2 : Rd > 0
then
    act1 : Rd := Rd - 1
end
Event resetC ≐
Status convergent
when
    grd1 : reset = TRUE
    grd2 : Rc > 0
then
    act1 : Rc := Rc - 1
end
Event reset ≐
refines reset
when
    grd1 : Ru = 0
    grd2 : Rd = 0
    grd3 : Rc = 0
    grd4 : reset = TRUE
then
    act1 : reset := FALSE
end
Event startReset ≐
Status convergent
when
    grd1 : Ru > 0 ∨ Rd > 0 ∨ Rc > 0
    grd2 : reset = FALSE
    grd3 : ud = 0
    grd4 : dc = 0
then
    act1 : reset := TRUE
end
VARIANT
2 * (1 - dc) + 1 - ud + Ru + Rd + Rc + 1 - b2n(reset)
END

```